depfile = $$replace(_PRO_FILE_PWD_, ([^/]+$), \\1/\\1_dependencies.pri)
exists($$depfile) {
    include($$depfile)
    isEmpty(PLUGIN_NAME): \
        error("$$basename(depfile) does not define PLUGIN_NAME.")
} else {
    isEmpty(PLUGIN_NAME): \
        error("PLUGIN_NAME is empty. Maybe you meant to create $$basename(depfile)?")
}

TARGET = $$PLUGIN_NAME

include(basic.pri)

INCLUDEPATH += $$OUT_PWD

TEMPLATE = lib
CONFIG += plugin
DESTDIR = $$APP_PLUGIN_PATH

TARGET = $$qtLibraryTargetName($$TARGET)

LIBS *= -l$$qtLibraryName(ExtensionSystem)
